

```

/*Exercise 1-Reading 2002 NAMCS public use data*/
/*Read NAMCS data & create a SAS dataset called namtest1 */

filename nam02 'c:\DUC_04\nam02'; /*data file*/

filename nam02inp 'c:\DUC_04\nam02inp.txt'; /*must have*/
filename nam02for 'c:\DUC_04\nam02for.txt'; /*variable labels*/
filename nam02lab 'c:\DUC_04\nam02lab.txt'; /*create var labels & assign
formats*/

%inc nam02for;

data namtest1;
infile nam02 missover lrecl=999;
%inc nam02inp;
%inc nam02lab;

patwt2=patwt/1000;
keep phycode patcode ager sex diag13d patwt patwt2 cpsum cstratm med1-med6
specr timemd drug1c11-drug1c13 drug2c11-drug2c13 drug3c11-drug3c13
drug4c11-drug4c13 drug5c11-drug5c13 drug6c11-drug6c13; /*35 variables*/

proc print data=namtest1 (obs=19);
title '2002 NAMCS Visits-Selected Variables';
run;
*****;
/* Exercise 2-Simple Frequencies, unweighted & weighted-SAS*/
/* Estimate the number of office visits by sex & age */

proc freq data=namtest1;
tables sex*ager;
title 'Unweighted 2002 NAMCS Visits: Sex by Age';
run;

proc freq data=namtest1;
tables sex*ager /list;
weight patwt2;
title 'Exercise 2-Weighted 2002 NAMCS Visits: Sex by Age';
run;
*****;
/*Exercise 3a-Simple Frequencies with standard errors-SAS Surveymeans*/
/*Estimate the number of office visits by sex & age with SEs */

data namtest2;
set namtest1;
all=1; /*dummy variable to produce total percent*/
proc surveymeans data=namtest2 nobn sum stderr mean;
domain sex; /*grouping*/
class all ager;
cluster cpsum;
strata cstratm;
var all ager;
weight patwt2;
TITLE 'Exercise 3a-2002 NAMCS Visits-Sex by Age (SAS Surveymeans)';
run;
*****;

```

```

/*Exercise 3b-Simple Frequencies with standard errors-SUDAAN */
/*Estimate the number of office visits by sex & age with SEs*/

proc sort data=namtest1;
by cstratm cpsum;
proc crosstab data=namtest1 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt2;
subgroup sex ager;
levels 2 6;
tables sex sex*ager;
print wsum sewgt totper setot nsum /
wsumfmt=f8.1 sewgtfmt=f8.1 colperfmt=f8.1 secolfmt=f8.1 nsumfmt=f8.
style=nchs;
Title "Exercise 3b-2002 NAMCS Visits-Sex & Sex by Age (SUDAAN)";
run;
*****;
/*Exercise 4a-Transforming data-SAS Surveymeans */
/*Breakdown of primary dx-listed asthma visits by sex*/

data namtest3;
set namtest1;

if DIAG13D='493' then asthma=1;
else asthma=2;

all=1;
proc surveymeans data=namtest3 nobs sum stderr mean;
domain asthma; /*grouping var*/
class all sex;
cluster cpsum;
strata cstratm;
var all sex;
weight patwt2;
TITLE "Exercise 4a-2002 NAMCS Asthma Visits by Sex (SAS Surveymeans)";
run;
*****;
/*Exercise 4b-Transforming data-SUDAAN */
/*Breakdown of primary dx-listed asthma visits by sex & age*/

data namtest4;
set namtest1;
if DIAG13D='493' then asthma=1;
else asthma=2;

proc sort data=namtest4;
by cstratm cpsum;
proc crosstab data=namtest4 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt2;
subpopn asthma=1; /*new part*/
subgroup sex ager;
levels 2 6;
tables sex*ager;
print wsum sewgt totper setot rowper serow nsum /
wsumfmt=f10.3 sewgtfmt=f8.3 totperfmt=f8.1
rowperfmt=f8.1 serowfmt=f8.1 nsumfmt=f8. style=nchs;

```

```

Title "Exercise 4b-2002 NAMCS Asthma Visits by Sex & Age (SUDAAN)";
run;
*****;
/*Exercise 5a-Creating visit rates-SAS Surveymeans*/
/*Estimate number of visits per 100 males & females*/

data namtest5;
set namtest1;
if sex=1 then sexwt=(patwt/145003609)*100;
else if sex=2 then sexwt=(patwt/138082547)*100;

proc surveymeans data=namtest5 nobs sum stderr mean;
  class sex;
  cluster cpsum;
  strata cstratm;
  var sex;
  weight sexwt;
Title "Exercise 5a-2002 NAMCS Visit Rates by Sex (SAS Surveymeans)";
run;
*****;
/*Exercise 5b-Creating Visit Rates-SUDAAN */
/*Estimate number of visits per 100 males & females*/

data namtest6;
set namtest1;
if sex=1 then sexwt=(patwt/145003609)*100;
else if sex=2 then sexwt=(patwt/138082547)*100;

proc sort data=namtest6;
by cstratm cpsum;
proc crosstab data=namtest6 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight sexwt;
subgroup sex;
levels 2;
tables sex;
print wsum sewgt nsum /
wsumfmt=f10.1 sewgtfmt=f8.1 nsumfmt=f8.
style=nchs;
Title "Exercise 5b-2002 NAMCS Visit Rates by Sex (SUDAAN)";
run;
*****;
/*Exercise 6a-Estimate total number of drug mentions-SAS Surveymeans*/

data namtest7;
set namtest1;
*****;
total=0;
*****;
array meds(6) med1-med6;
do j=1 to 6;
  if meds(j)ne '90000' then total=total+1; /*90000=no entry made*/
end;
proc surveymeans data=namtest7 nobs sum stderr;
  *class total; /*class n/a because total is continuous*/
  cluster cpsum;
  strata cstratm;

```

```

var total;
weight patwt2;
Title "Exercise 6a-2002 NAMCS Drug Mentions (SAS Surveymeans)";
run;
*****;
/*Exercise 6b-Estimate total number of drug mentions-SUDAAN*/

data namtest8;
set namtest1;
*****;
total=0;
*****;
array meds(6) med1-med6;
do j=1 to 6;
if meds(j)ne '90000' then total=total+1; /*90000=no entry made*/
end;
proc sort data=namtest8;
by cstratm cpsum;
proc descript data=namtest8 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt2;
var total;
print total settotal nsum /
totalfmt=f13.3 settotalfmt=f10.3 nsumfmt=f8.
style=nchs;
Title "Exercise 6b-2002 NAMCS Drug Mentions (SUDAAN)";
run;
*****;
/*Exercise 7a-Number of antidepressant drug mentions by
therapeutic class-SAS Surveymeans */

data namtest9;
set namtest1;
*****;
antidep=0;
*****;
array dclass(18)
drug1cl1-drug1cl3
drug2cl1-drug2cl3
drug3cl1-drug3cl3
drug4cl1-drug4cl3
drug5cl1-drug5cl3
drug6cl1-drug6cl3;
DO J=1 TO 18;

if dclass(j)='0630' then antidep=antidep +1;
end;
proc surveymeans data=namtest9 nobsum sum mean;
cluster cpsum;
strata cstratm;
var antidep;
weight patwt2;
Title "Exercise 7a-2002 NAMCS Antidepressant Drug Mentions
by therapeutic class (SAS Surveymeans)";
run;
*****;
/*Exercise 7b-Number of antidepressant drug mentions-SUDAAN */

```

```

data namtst10;
set namtest1;
*****;
antidep=0;
*****;
array dclass(18)
drug1cl1-drug1cl3
  drug2cl1-drug2cl3
  drug3cl1-drug3cl3
  drug4cl1-drug4cl3
  drug5cl1-drug5cl3
  drug6cl1-drug6cl3;
DO J=1 TO 18;

if dclass(j)= '0630' then antidep=antidep +1;
end;

proc sort data=namtst10;
by cstratm cpsum;
proc descript data=namtst10 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt2;
var antidep;
print total settotal nsum /
totalfmt=f10.3 settotalfmt=f8.3 nsumfmt=f8.
style=nchs;
Title "Exercise 7b-2002 NAMCS Antidepressant Drug Mentions
by therapeutic class (SUDAAN)";
run;
*****;
/*Exercise 8a-Mean number of minutes spent with physician-SAS Surveymeans */

data namtst11;
set namtest1;

if timemd=0 then timemd=.; /*used to delete timemd=0 visits*/

proc surveymean data=namtst11 nobs sum stderr mean;
  domain specr; /*grouping var*/
  class specr;
  cluster cpsum;
  strata cstratm;
  var timemd;
  weight patwt2;
Title "Exercise 8a-2002 NAMCS Time Spent with Physician by Specialty
(SAS Procsurveymean)";
run;
*****;
/*Exercise 8b-Mean number of minutes spent with physician-SUDAAN */

data namtst12;
set namtest1;

if timemd=0 then timemd=.; /*used to delete timemd=0 visits*/

proc sort data=namtst12;

```

```
by cstratm cpsum;
proc descript data=namtst12 filetype=sas DESIGN = WR;
NEST cstratm cpsum;
weight patwt2;
subgroup specr;
levels 15;
var timemd;
print mean semean nsum /
totalfmt=f10.3 settotalfmt=f8.3 nsumfmt=f8.
style=nchs;
Title "Exercise 8b-2002 NAMCS Time Spent with Physician by Specialty
(SUDAAN)";
run;
*****;
```